# Maintainer: Alexander F. Rødseth <xyproto@archlinux.org>
# Contributor: Douglas Soares de Andrade <dsa@aur.archlinux.org>
# Contributor: Magnus Jonsson <jmagnusj@gmail.com>

pkgname=qgo
pkgver=2.1.1
pkgrel=1
pkgdesc='Go client and full featured SGF editor'
arch=(loong64 x86_64)
url='https://github.com/johndaniels/qgo'
license=(GPL)
depends=(libsm qt5-multimedia)
makedepends=(git qt5-tools)
options=(!emptydirs)
source=("git+$url#commit=4b39e8270d366026be984b8bdac23348f1572e39") # tag: 2.1.1
b2sums=('SKIP')

build() {
  cd $pkgname/src
  qmake
  make CXXFLAGS+='-fPIC -w'
}

package() {
  make -C $pkgname/src INSTALL_ROOT="$pkgdir" install
}
